

## Postestimation commands

The following postestimation commands are available after `xtlogit`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of parameters
* <code>estat ic</code>	Akaike's, consistent Akaike's, corrected Akaike's, and Schwarz's Bayesian information criteria (AIC, CAIC, AICc, and BIC, respectively)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
† <code>forecast</code>	dynamic forecasts and simulations
<code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of parameters
* <code>lrtest</code>	likelihood-ratio test
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of parameters
<code>predict</code>	linear predictions and their SEs, probabilities
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of parameters
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

\*`estat ic` and `lrtest` are not appropriate after `xtlogit, pa`.

†`forecast` is not appropriate with `mi` estimation results or after `xtlogit, fe`.

# predict

## Description for predict

predict creates a new variable containing predictions such as linear predictions, probabilities, standard errors, and the equation-level score.

## Menu for predict

Statistics > Postestimation

## Syntax for predict

Random-effects model

```
predict [type] newvar [if] [in] [ , RE_statistic nooffset ]
```

Fixed-effects model

```
predict [type] newvar [if] [in] [ , FE_statistic nooffset ]
```

Population-averaged model

```
predict [type] newvar [if] [in] [ , PA_statistic nooffset ]
```

RE_statistic	Description
Main	
xb	linear prediction; the default
pr	marginal probability of a positive outcome
pu0	probability of a positive outcome assuming that the random effect is zero
stdp	standard error of the linear prediction

FE_statistic	Description
Main	
pc1	predicted probability of a positive outcome conditional on one positive outcome within group; the default
pu0	probability of a positive outcome assuming that the fixed effect is zero
xb	linear prediction
stdp	standard error of the linear prediction
score	first derivative of the log likelihood with respect to $\mathbf{x}_{it}\beta$

The predicted probability for the fixed-effects model is conditional on there being only one outcome per group. See [R] [clogit](#) for details.

<i>PA_statistic</i>	Description
Main	
<code>mu</code>	predicted probability of <i>depvar</i> ; considers the <code>offset()</code>
<code>rate</code>	predicted probability of <i>depvar</i>
<code>xb</code>	linear prediction
<code>stdp</code>	standard error of the linear prediction
<code>score</code>	first derivative of the log likelihood with respect to $\mathbf{x}_{it}\beta$

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

## Options for predict

### Main

`xb` calculates the linear prediction. This is the default for the random-effects model.

`pc1` calculates the predicted probability of a positive outcome conditional on one positive outcome within group. This is the default for the fixed-effects model.

`mu` and `rate` both calculate the predicted probability of *depvar*. `mu` takes into account the `offset()`, and `rate` ignores those adjustments. `mu` and `rate` are equivalent if you did not specify `offset()`. `mu` is the default for the population-averaged model.

`pr` calculates the probability of a positive outcome that is marginal with respect to the random effect, which means that the probability is calculated by integrating the prediction function with respect to the random effect over its entire support.

`pu0` calculates the probability of a positive outcome, assuming that the fixed or random effect for that observation's panel is zero ( $\nu_i = 0$ ). This may not be similar to the proportion of observed outcomes in the group.

`stdp` calculates the standard error of the linear prediction.

`nooffset` is relevant only if you specified `offset(varname)` for `xtlogit`. This option modifies the calculations made by `predict` so that they ignore the offset variable; the linear prediction is treated as  $\mathbf{x}_{it}\beta$  rather than  $\mathbf{x}_{it}\beta + \text{offset}_{it}$ .

`score` calculates the equation-level score,  $u_{it} = \partial \ln L(\mathbf{x}_{it}\beta) / \partial (\mathbf{x}_{it}\beta)$ .

# margins

## Description for margins

margins estimates margins of response for linear predictions and probabilities.

## Menu for margins

Statistics > Postestimation

## Syntax for margins

```
margins [marginlist] [ , options ]
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

### Random-effects model

statistic	Description
pr	marginal probability of a positive outcome; the default
pu0	probability of a positive outcome assuming that the random effect is zero
xb	linear prediction
stdp	not allowed with margins

### Fixed-effects model

statistic	Description
pu0	probability of a positive outcome assuming that the fixed effect is zero; the default
xb	linear prediction
pc1	not allowed with margins
stdp	not allowed with margins
<u>s</u> core	not allowed with margins

### Population-averaged model

statistic	Description
mu	probability of <i>depvar</i> ; considers the <code>offset()</code>
rate	probability of <i>depvar</i>
xb	linear prediction
stdp	not allowed with margins
<u>s</u> core	not allowed with margins

Statistics not allowed with margins are functions of stochastic quantities other than  $e(b)$ .  
 For the full syntax, see [\[R\] margins](#).

## Remarks and examples

### ▷ Example 1: Conducting hypothesis tests

In [example 1](#) of [\[XT\] xtlogit](#), we fit a random-effects model of union status on the person's age and level of schooling, whether she lived in an urban area, and whether she lived in the south. In fact, we included the full interaction between south and year to capture both the overall effect of residing in the south and a separate time-trend for southerners. To test whether residing in the south affects union status, we must determine whether `1.south` and `south#c.year` are jointly significant. First, we refit our model, store the estimation results for later use, and use `test` to conduct a Wald test of the joint significance of those two variables' parameters:

```
. use https://www.stata-press.com/data/r19/union
(NLS Women 14-24 in 1968)
. xtlogit union age grade not_smsa south##c.year
(output omitted)
. estimates store fullmodel
. test 1.south 1.south#c.year
( 1) [union]1.south = 0
( 2) [union]1.south#c.year = 0
      chi2( 2) = 143.93
      Prob > chi2 = 0.0000
```

The test statistic is clearly significant, so we reject the null hypothesis that the coefficients are jointly zero and conclude that living in the south does significantly affect union status.

We can also test our hypothesis with a likelihood-ratio test. Here we fit the model without `south##c.year` and then call `lrtest` to compare this restricted model to the full model:

```
. xtlogit union age grade not_smsa
(output omitted)
. lrtest fullmodel .
Likelihood-ratio test
Assumption: . nested within fullmodel
      LR chi2(3) = 146.55
      Prob > chi2 = 0.0000
```

These results confirm our finding that living in the south affects union status.



## Also see

[\[XT\] xtlogit](#) — Fixed-effects, random-effects, and population-averaged logit models

[\[U\] 20 Estimation and postestimation commands](#)

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).